Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 20
Triple Integrals

Initializations

```
> restart;
  with(plots):
  with(student):
>
```

20.1 Triple Integrals

With a double integral one integrates a function of two variables over a region in the *xy*-plane. With a triple integral one integrates a function of three variables over a region in 3-space.

Examples

Example 20.1.1

Integrate the function $f(x, y, z) = 2xy + 5yz^2$ over the region in the first octant bounded by the coordinate planes and the plane $V: x + 3y + 4z = 7$.

Solution
Code the function *f* and the plane *V*.

```
> f:=(x, y, z)->2*x*y+5*y*z^2;
  V:=x+3*y+4*z=7;
```

$$f := (x, y, z) \rightarrow 2yx + 5yz^2$$

$$V := x + 3y + 4z = 7 \qquad \textbf{(2.1.1.1)}$$

Observe that the region of integration is a pyramid bounded by the coordinate planes and *V*. It is VERY important that you understand the topology of the configuration, just by looking at the formulas. Since the region of integration is bounded below by a triangle in the *xy* plane and above by the slanted plane *V* it makes sense to integrate over z first. In order to determine the order of the *xy* integration, we determine the bound of the footprint of the integration region and make a sketch.

```
> bnd_1:=x=0;
  bnd_2:=y=0;
  bnd_3:=subs(z=0, V);
```

$$bnd\_1 := x = 0$$

$$bnd\_2 := y = 0$$

$$bnd\_3 := x + 3y = 7 \qquad \textbf{(2.1.1.2)}$$

```
> implicitplot({seq(bnd_||k, k=1..3)}, x=-1..8, y=-1..3,
```

```
axes=boxed);
```



If we first integrate over z, then over y and finally over x, our limits will look as follows:

```
> zmin:=0;
  zmax:=solve(V, z);
```

$$zmin := 0$$

$$zmax := -\frac{1}{4}x - \frac{3}{4}y + \frac{7}{4} \quad \textbf{(2.1.1.3)}$$

```
> ymin:=0;
  ymax:=solve(bnd_3, y);
```

$$ymin := 0$$

$$ymax := -\frac{1}{3}x + \frac{7}{3} \quad \textbf{(2.1.1.4)}$$

```
> xmin:=0;
  xmax:=solve(subs(y=0, bnd_3), x);
```

$$xmin := 0$$

$$xmax := 7 \quad \textbf{(2.1.1.5)}$$

```
> e1:=Tripleint(f(x, y, z), z=zmin..zmax, y=ymin..ymax, x=
  xmin..xmax);
```

$$e1 := \int_0^7 \int_0^{-\frac{1}{3}x+\frac{7}{3}} \int_0^{-\frac{1}{4}x-\frac{3}{4}y+\frac{7}{4}} \left(2\,y\,x + 5\,y\,z^2\right) \mathrm{d}z\,\mathrm{d}y\,\mathrm{d}x \quad \textbf{(2.1.1.6)}$$

```
> answer:=value(e1);
  evalf(%);
```

$$answer := \frac{2201717}{207360}$$

$$10.61784819 \quad \textbf{(2.1.1.7)}$$

```
>
```

Example 29.1.2

Evaluate $\iiint_D \sqrt{2\,y^2 + 2\,z^2 + 3}\; dV$, where D is the region in three space bounded by the

paraboloid S: $x = y^2 + z^2$ and the plane T: $x = 5$.

Solution
First observe that the paraboloid S is centered about the x-axes. The **implicitplot3d** command can be used to obtain a crude picture of the integration region.

```
> S:=x=y^2+z^2;
  T:=x=5;
  f:=(x, y, z)->sqrt(2*y^2+2*z^2+3);
```

$$S := x = y^2 + z^2$$

$$T := x = 5$$

$$f := (x, y, z) \rightarrow \sqrt{2\,y^2 + 2\,z^2 + 3} \quad \textbf{(2.1.2.1)}$$

```
> p1:=implicitplot3d(S, x=0..6, y=-3..3, z=-3..3, style=
  patch):
> p2:=plot3d([5, y, z], y=-3..3, z=-3..3, style=patch,
  color=red):
> display([p1, p2], orientation=[-64, 82], labels=[x, y, z]
  , scaling=constrained, axes=boxed);
```



The computational complexity of the **implicitplot3d** routine clearly takes its toll. We can obtain much better results by revolving the curve x = y^2 about the x-axes. That approach allows for a parametric representation of the surface, which is computationally much more efficient. With a parametric representation, we can even open-up the solid and look into the integration region.

```
> p3:=plot3d([y^2, y*cos(u), y*sin(u)], u=Pi/3..2*Pi, y=0.
  .2.5, style=patch):
  display([p2, p3], orientation=[117, 75], labels=[x, y, z]
  , scaling=constrained, axes=boxed);
```



It seems clear that the quickest way to solve the problem is first to integrate over x, between the plane T and the paraboloid S and then use an integration in polar coordinates over the base of the paraboloid. Observe that we use polar coordinates in a plane parallel to the yz-plane, therefore $y = r\cos\theta$ and $z = r\sin\theta$., while $dA = r\,dy\,dz$.

```
> S;
  T;
  polar:={y=r*cos(theta), z=r*sin(theta)};
```

$$x = y^2 + z^2$$

$$x = 5$$

$$\text{polar} := \{y = r\cos(\theta), z = r\sin(\theta)\} \qquad \textbf{(2.1.2.2)}$$

```
> base:=subs(T, S);
```

$$base := 5 = y^2 + z^2 \qquad \textbf{(2.1.2.3)}$$

```
> xmin:=solve(S, x);
  xmax:=5;
```

$$xmin := y^2 + z^2$$

$$xmax := 5 \qquad \textbf{(2.1.2.4)}$$

Warning!!! Do not forget to convert **xmin** to polar coordinates.

```
> xmin:=simplify(subs(polar, xmin));
```

$$xmin := r^2 \qquad \textbf{(2.1.2.5)}$$

```
> rmin:=0;
  rmax:=sqrt(5);
```

$$rmin := 0$$

$$rmax := \sqrt{5} \qquad \textbf{(2.1.2.6)}$$

```
> thetamin:=0;
  thetamax:=2*Pi;
```

$$thetamin := 0$$

(2.1.2.7)

$$thetamax := 2\pi \quad (2.1.2.7)$$

```
> e1:=Tripleint(subs(polar, f(x, y, z))*r, x=xmin..xmax, r=
  rmin..rmax, theta=thetamin..thetamax);
```

$$e1 := \int_0^{2\pi}\int_0^{\sqrt{5}}\int_{r^2}^{5} \sqrt{2 r^2 \cos(\theta)^2 + 2 r^2 \sin(\theta)^2 + 3}\ r \,dx\,dr\,d\theta \quad (2.1.2.8)$$

```
> e2:=simplify(e1);
```

$$e2 := \left(\int_0^{\sqrt{5}} \sqrt{3 + 2 r^2}\ r \left(\int_{r^2}^{5} 1\,dx\right) dr\right)\left(\int_0^{2\pi} 1\,d\theta\right) \quad (2.1.2.9)$$

```
> answer:=combine(simplify(value(e2)), radical);
  evalf(%);
```

$$answer := \frac{1}{45}\sqrt{3}\left(169\sqrt{39} - 252\right)\pi$$

$$97.14765770 \quad (2.1.2.10)$$

```
>
```